Была старая обработка по загрузке реализации в 1с БУХ 3. Теперь при последнем обновлении на (3.0.121.25) Выдает такую ошибку:
Не удалось загрузить данные! {Общий модуль.РеализацияТоваровУслугФормы.Модуль(442)}:
Поле объекта не обнаружено (ГосконтрактЕИС)
Подскажите как можно эту исправить в обработке, что там вообще в этом параметре должно быть указанно или как его пропустить.
---
в обработке ДанныеОбъекта это кто? Реквизит или таблица\масссив\соответствие?
---
имеется в виду программно формируемый таблица\масссив\соответствие или таблица на форме?
---
Все разобрался)) Спасибо! Добавил ГосконтрактЕИС сюда:

```1c
    ДанныеОбъекта = Новый Структура("Ссылка, Дата, ВидОперации, Организация, ДеятельностьНаПатенте,
    |Склад, ТипЦен, ВалютаДокумента, КурсВзаиморасчетов, КратностьВзаиморасчетов,
    |ГосконтрактЕИС,СуммаВключаетНДС, ДоговорКонтрагента,
    |ЭтоКомиссия, Реализация, ДокументБезНДС");
```
---
Видимо госконтракт должен быть))
Напиши строку эту и как этот модуль вызывается из обработки.
---
Эту строку просит как раз таки не обработка а 1С. Видимо в обновлении какие то изменения внесли.
---
*
---
строку скинь 442ю
---
```1c
    ЭлектронноеАктированиеБП.ДополнитьДанныеСтрокиСведениямиПоГосконтракту(СтрокаТабличнойЧасти, ДанныеОбъекта.ГосконтрактЕИС);
```
---
ГосконтрактЕИС в договоре есть такой реквизит, ДанныеОбъекта это к документу относится? В документе ЭтапГосконтрактаЕИС и ИдентификаторСтрокиГосконтрактаЕИС.
---
Везде читаю что данный функционал включен только в версии КОРП а у нас ПРОФ.
Скорее всего ДанныеОбъекта это к документу реализации и относится. Вижу в нем эти реквизиты. Как их допустим оставить пустыми чтобы данные загрузились?
---
Не, вру в договоре ГосударственныйКонтракт реквизит. Смотри ДанныеОбъекта откуда берет.
---
Если у тебя ЭлектронноеАктированиеБП.ДополнитьДанныеСтрокиСведениямиПоГосконтракту( из твоей обработки не вызывается - смотри что вызывается - и откуда вызывается этот метод, как то же туда передается ДанныеОбъекта.
---
в обработке так:

```1c
    Процедура УслугиНоменклатураПриИзмененииНаСервере(СтрокаТабличнойЧасти, Знач ДанныеОбъекта, Знач СчетаУчетаКЗаполнению)

    РеализацияТоваровУслугФормы.УслугиНоменклатураПриИзмененииНаСервере(СтрокаТабличнойЧасти, ДанныеОбъекта, СчетаУчетаКЗаполнению);

    ЗаполнитьДобавленныеКолонкиСтрокиТаблицыУслуги(СтрокаТабличнойЧасти, ДанныеОбъекта);

    КонецПроцедуры
```

В 1с было так:

```1c
    Процедура УслугиНоменклатураПриИзмененииНаСервере(СтрокаТабличнойЧасти, Знач ДанныеОбъекта, Знач СчетаУчетаКЗаполнению) Экспорт

    Результат = СчетаУчетаВДокументах.ЗаполнитьРеквизитыПриИзменении(
    Документы.РеализацияТоваровУслуг,
    СчетаУчетаКЗаполнению,
    ДанныеОбъекта,
    "Услуги",
    СтрокаТабличнойЧасти);

    ЗаполнитьЗначенияСвойств(СтрокаТабличнойЧасти, Результат);

    Если Не ЗначениеЗаполнено(ДанныеОбъекта.ТипЦен) Тогда
    ДанныеОбъекта.Вставить("СпособЗаполненияЦены", Перечисления.СпособыЗаполненияЦен.ПоПродажнымЦенам);
    КонецЕсли;

    СведенияОНоменклатуре = БухгалтерскийУчетПереопределяемый.ПолучитьСведенияОНоменклатуре(
    СтрокаТабличнойЧасти.Номенклатура, ДанныеОбъекта, Ложь);
    Если СведенияОНоменклатуре = Неопределено Тогда
    Возврат;
    КонецЕсли;

    СтрокаТабличнойЧасти.Содержание = СведенияОНоменклатуре.НаименованиеПолное;
    СтрокаТабличнойЧасти.Цена = СведенияОНоменклатуре.Цена;
    СтрокаТабличнойЧасти.СтавкаНДС = СведенияОНоменклатуре.СтавкаНДС;

    ОбработкаТабличныхЧастейКлиентСервер.РассчитатьСуммуТабЧасти(СтрокаТабличнойЧасти, 1);
    ОбработкаТабличныхЧастейКлиентСервер.РассчитатьСуммуНДСТабЧасти(СтрокаТабличнойЧасти, ДанныеОбъекта.СуммаВключаетНДС);

    КонецПроцедуры
```

 А стало так:

```1c
    Процедура УслугиНоменклатураПриИзмененииНаСервере(СтрокаТабличнойЧасти, Знач ДанныеОбъекта, Знач СчетаУчетаКЗаполнению) Экспорт

    Результат = СчетаУчетаВДокументах.ЗаполнитьРеквизитыПриИзменении(
    Документы.РеализацияТоваровУслуг,
    СчетаУчетаКЗаполнению,
    ДанныеОбъекта,
    "Услуги",
    СтрокаТабличнойЧасти);

    ЗаполнитьЗначенияСвойств(СтрокаТабличнойЧасти, Результат);

    Если Не ЗначениеЗаполнено(ДанныеОбъекта.ТипЦен) Тогда
    ДанныеОбъекта.Вставить("СпособЗаполненияЦены", Перечисления.СпособыЗаполненияЦен.ПоПродажнымЦенам);
    КонецЕсли;

    СведенияОНоменклатуре = БухгалтерскийУчетПереопределяемый.ПолучитьСведенияОНоменклатуре(
    СтрокаТабличнойЧасти.Номенклатура, ДанныеОбъекта, Ложь);
    Если СведенияОНоменклатуре = Неопределено Тогда
    Возврат;
    КонецЕсли;

    СтрокаТабличнойЧасти.Содержание = СведенияОНоменклатуре.НаименованиеПолное;
    СтрокаТабличнойЧасти.Цена = СведенияОНоменклатуре.Цена;
    СтрокаТабличнойЧасти.СтавкаНДС = СведенияОНоменклатуре.СтавкаНДС;

    ЭлектронноеАктированиеБП.ДополнитьДанныеСтрокиСведениямиПоГосконтракту(СтрокаТабличнойЧасти, ДанныеОбъекта.ГосконтрактЕИС);

    ОбработкаТабличныхЧастейКлиентСервер.РассчитатьСуммуТабЧасти(СтрокаТабличнойЧасти, 1);
    ОбработкаТабличныхЧастейКлиентСервер.РассчитатьСуммуНДСТабЧасти(СтрокаТабличнойЧасти, ДанныеОбъекта.СуммаВключаетНДС);

    КонецПроцедуры
```
---
там ссылка на СправочникСсылка.ГосКонтракты. У нас этот справочник даже неактивен
---
в обработке ДанныеОбъекта это кто? Реквизит или таблица\масссив\соответствие?
---
```1c
    Функция ОбработатьСтрокуНоменклатуры(Объект,Знач СтрокаТабличнойЧасти)
    // Получим общие параметры обработки для реквизитов документа
    //ПараметрыОбработки = РеализацияТоваровУслугФормыКлиент.ПодготовитьПараметрыОбработкиУслугиНоменклатураПриИзменении(
    // ЭтаФорма, ТекущиеДанные);
    СтрокаТабличнойЧасти=СтрокаТабличнойЧастиВСтруктуру(СтрокаТабличнойЧасти);

    //Объект = Форма.Объект;
    ЭтоКомиссия = Ложь;

    ДанныеСтрокиТаблицы = Новый Структура("Номенклатура, Содержание, Количество,
    |Цена, Сумма, СтавкаНДС, СуммаНДС");
    ЗаполнитьЗначенияСвойств(ДанныеСтрокиТаблицы, СтрокаТабличнойЧасти);

    ДанныеОбъекта = Новый Структура("Ссылка, Дата, ВидОперации, Организация, ДеятельностьНаПатенте,
    |Склад, ТипЦен, ВалютаДокумента, КурсВзаиморасчетов, КратностьВзаиморасчетов,
    |СуммаВключаетНДС, ДоговорКонтрагента,
    |ЭтоКомиссия, Реализация, ДокументБезНДС");
    ЗаполнитьЗначенияСвойств(ДанныеОбъекта, Объект);
    ДанныеОбъекта.ЭтоКомиссия = ЭтоКомиссия;
    ДанныеОбъекта.Реализация = Истина;

    ПараметрыЗаполненияСчетовУчета = РеализацияТоваровУслугФормыКлиентСервер.НачатьЗаполнениеСчетовУчета(
    "Услуги.Номенклатура",
    Объект,
    СтрокаТабличнойЧасти,
    ДанныеОбъекта,
    ДанныеСтрокиТаблицы);

    ПараметрыОбработки = Новый Структура();
    ПараметрыОбработки.Вставить("ДанныеСтрокиТаблицы", ДанныеСтрокиТаблицы);
    ПараметрыОбработки.Вставить("ДанныеОбъекта", ДанныеОбъекта);
    ПараметрыОбработки.Вставить("СчетаУчетаКЗаполнению", ПараметрыЗаполненияСчетовУчета.КЗаполнению);

    // Дополнительные поля, требующиеся для заполнения добавленных колонок табличного поля текущей формы.
    ПараметрыОбработки.ДанныеСтрокиТаблицы.Вставить("Всего", 0);
    ПараметрыОбработки.ДанныеСтрокиТаблицы.Вставить("АналитикаУчета", ");
    ПараметрыОбработки.ДанныеСтрокиТаблицы.Вставить("АналитикаУчетаЗаполнена", Ложь);

    ПараметрыОбработки.ДанныеСтрокиТаблицы.Вставить("СчетДоходов");
    ПараметрыОбработки.ДанныеСтрокиТаблицы.Вставить("Субконто");
    ПараметрыОбработки.ДанныеСтрокиТаблицы.Вставить("СчетУчетаНДСПоРеализации");
    ПараметрыОбработки.ДанныеСтрокиТаблицы.Вставить("СчетРасходов");

    //
    ЗаполнитьПараметрыОбъектаДляЗаполненияДобавленныхКолонок(Объект, ПараметрыОбработки.ДанныеОбъекта);

    УслугиНоменклатураПриИзмененииНаСервере(
    ПараметрыОбработки.ДанныеСтрокиТаблицы,
    ПараметрыОбработки.ДанныеОбъекта,
    ПараметрыОбработки.СчетаУчетаКЗаполнению);

    ЗаполнитьЗначенияСвойств(СтрокаТабличнойЧасти, ПараметрыОбработки.ДанныеСтрокиТаблицы);

    Возврат(СтрокаТабличнойЧасти);
    КонецФункции // ()

    Процедура УслугиНоменклатураПриИзмененииНаСервере(СтрокаТабличнойЧасти, Знач ДанныеОбъекта, Знач СчетаУчетаКЗаполнению)

    РеализацияТоваровУслугФормы.УслугиНоменклатураПриИзмененииНаСервере(СтрокаТабличнойЧасти, ДанныеОбъекта, СчетаУчетаКЗаполнению);

    ЗаполнитьДобавленныеКолонкиСтрокиТаблицыУслуги(СтрокаТабличнойЧасти, ДанныеОбъекта);

    КонецПроцедуры
```
---
имеется в виду программно формируемый таблица\масссив\соответствие или таблица на форме?
---
Все разобрался)) Спасибо! Добавил ГосконтрактЕИС сюда:

```1c
    ДанныеОбъекта = Новый Структура("Ссылка, Дата, ВидОперации, Организация, ДеятельностьНаПатенте,
    |Склад, ТипЦен, ВалютаДокумента, КурсВзаиморасчетов, КратностьВзаиморасчетов,
    |ГосконтрактЕИС,СуммаВключаетНДС, ДоговорКонтрагента,
    |ЭтоКомиссия, Реализация, ДокументБезНДС");
```